Maskinarkitektur og operativsystemer

OPGAVER
5.2.a.2



5.2.aPRIVATE 


8086 assemblerprogrammer


Opgaver

Bjørk Busch
Opgave 1.
Omsæt følgende programstump i pascal-notation til 8086 assem​bler:


Result := (Tal1 + Tal2) * 3

Alle variable er af typen integer. Find selv på testværdier.

Multiplikation skal klares uden brug af multiplikations-in​struktioner og uden løkke (konstant 3 gange).

Opgave 2.
Udbyg ovenstående programstump med følgende (efter beregning i opgave 1):


IF (Result < 0) then



Result := Result + 100


ELSE (IF Result = 0) then



Result := 20


ELSE



Result := Result - 100;

Opgave 3.
Omsæt følgende programstump i pascal-notation til 8086 assem​bler:


SUM := 0;


FOR i := 1 to 4 DO BEGIN



SUM := SUM + ITAB[i];


END;

Tabellen kan defineres således i assembler:


ITAB

DW

11,21,31,41

TIPS!
Der skal anvendes base-/index-register som ved ek​sempel fra MOV. Denne adres​sering kan også anvendes i forbindelse med ADD.

Opgave 4.
Givet er følgende datafelter:

INAVN

DB
'EDB-SKOLEN','$'

UNAVN

DB
30 DUP (' ')




DB
13,10,'$'

Det er med vilje, at ascii-tegnene 13, 10 og $ kommer efter UNAVN og at der ikke påhæftes et symbolsk navn (symbolsk-adresse).

Du bedes udarbejde en assembler-rutine, der kan flytte teksten (excl. $) fra INAVN til UNAVN, idet der skal være et mellemrum på et tegn mellem hvert bogstav i UNAVN.

Opgave 5.
Skriv programmet fra opgave 4 så det kan oversættes med MASM, LINKES og aftestes.

Prøv af afvikle programmet under CODEVIEW.

Opgave 6.
Programmet fra opgave 5 skal nu udbygges, så de to tekster udskrives på skærmen (DOS funktionskald 9 - int 21h).

Offset-adressen på en variabel kan fås på følgende måde:


MOV
DX,Offset xxxxx


Hvor xxxxx er det symbolske navn.

Prøv programmet under CODEVIEW, samt direkte ude i DOS.

Opgave 7.
Lav en rutine, der kan optælle antallet af ascii-tegnet A i en tekstvaria​bel, idet variablen afsluttes med et binært nul.

Opgave 8.
Lav en procedure, der kan omsætte et binært tal (16 bit word) til ascii-format.

Proceduren skal hedde BIN2ASC og være at typen NEAR.

Proceduren skal kaldes med følgende parametre overført i registre:


Ax

Word med det binære tal, der skal konver​teres.


Cx

Antallet af cifre, der skal være i det kon​ver​terede tal (mindst 1 ciffer).


Ds:Di
Startadressen på en tekststreng med plads til tallet i asciiformat.

Eksempler på hvordan proceduren skal kunne bruges:




MOV

Ax,BinTal1




LEA

Di,AscTal1




MOV

Cx,2




CALL

BIN2ASC




LEA

Dx,Linie1




CALL

UDSKRIV




MOV

Ax,BinTal2




LEA

Di,AscTal2




MOV

Cx,3




CALL

BIN2ASC




LEA

Dx,Linie2




CALL

UDSKRIV

Rutinen UDSKRIV er defineret på følgende måde:


UDSKRIV
PROC

NEAR




PUSH

Ax

;Gem Ax på STAK




MOV

Ah,09h
;Kald DOS for




INT

21

;    udskrift




POP

Ax

;Restore Ax fra STAK




RET



;Returer


UDSKRIV
ENDP

I datasegmentet er følgende definitioner:


BinTal1
Dw

27


Linie1
Db

'Første tal var: '


AscTal1
Db

2 DUP (?)




Db

10,13,'$'

;Slut med linieskift


BinTal2
Dw

743


Linie2
Db

'Andet tal var: '


AscTal2
Db

3 DUP (?)




Db

10,13,'$'

;Slut med linieskift

Opgave 8 (fortsat).
Hjælp til program-algoritme:

-
Start med at gemme alle registre, der ændres i proce​duren, på stak​ken.

-
Sæt Di til at pege på det sidste ciffer i tallet (det mindst be​tydende - højre).

-
Ved heltalsdivision med 10 fås nu en kvotient og en rest. 123/10 giver 12 som kvotient og 3 som rest.

-
Resten som jo er et binært tal mellem 0 og 9 vil kunne stå i en byte og kan dirrekte omregnes til et ascii-tegn.


Find værdien i ascci-tabellen.

-
Det beregnede ascii-tegn flyttes nu på plads i lageret.


Ved at gentage division med 10 og det efterfølgende fås et ciffer fra højre af gangen, og dette skal så ske et antal gange svarede til hvad proceduren fik leveret i Cx.

-
Slut af med at restore alle de gemte registre fra stakken og returner til kaldende rutine.

Opgave 9.
Lav en procedure, der kan omsætte et tal i ascii-format til binært (16 bit word).

Proceduren skal hedde ASC2BIN og være at typen NEAR.

Proceduren skal kaldes med følgende parametre overført i registre:


Ds:Si
Startadressen på en tekststreng med tallet i ascii-format.


Cx

Antallet af cifre, der skal konverteres.


Ax

Resultatet (det konverterede tal).

Hjælp til program-algoritme:


BINTAL := 0;


FOR I:= 1 To ANTALCIFRE DO BEGIN



BINTAL := BINTAL * 10;



BINTAL := BINTAL + ( ORD(ASCTAL[I]) - ORD('0') );


END;

Opgave 10.
DATA‑SEGMENT
ADRESSE    HEX‑DUMP AF DATASEGMENT 

seg :ofs



FØR KØRSEL
2026:0000  05 00 0B 00 16 00 21 00‑2C 00 2D 00 00 00

seg :ofs



EFTER KØRSEL
2026:0000  05 00 0B 00 16 00 21 00‑2C 00 2D 00 9B 00

           ‑‑‑‑‑ ‑‑‑‑‑ ‑‑‑‑‑ ‑‑‑‑‑ ‑‑‑‑‑ ‑‑‑‑‑ ‑‑‑‑‑

           Antal ‑‑‑‑‑‑‑‑‑‑‑‑‑Itab‑‑‑‑‑‑‑‑‑‑‑‑ SUM

CODE‑SEGMENT
ADRESSE   MASKINKODE    SYMBOLSK KODE          kommentarer    

seg :ofs                                       (ikke DEBUG)   

2027:0000 E80600        CALL
0009             NB 0006 relativ

2027:0003 E80F00        CALL
0015             NB 000F relativ

2027:0006 E80600        CALL
000F             NB 0006 relativ

2027:0009 B82620        MOV
AX,2026     

2027:000C 8ED8          MOV
DS,AX       

2027:000E C3            RET
            

2027:000F B000          MOV
AL,00       

2027:0011 B44C          MOV
AH,4C       

2027:0013 CD21          INT
21          

2027:0015 33C0          XOR
AX,AX       

2027:0017 33F0          XOR
SI,SI       

2027:0019 8B0E0000      MOV
CX,[0000]   

2027:001D 83F900        CMP
CX,+00      

2027:0020 7E0A          JLE
002C             NB 0A rel. +10

2027:0022 03840200      ADD
AX,[SI+0002]

2027:0026 83C602        ADD
SI,+02      

2027:0029 49            DEC
CX          

2027:002A EBF4          JMP
0020             NB F4 rel. -12

2027:002C A30C00        MOV
[000C],AX   

2027:002F C3            RET
            

a)
Ovenstående viser et assemblerprogram udskrevet i DEBUG-format.


Foretag sammenligning med programmet, som det så ud i kildetekst (på næste side).

b)
Foretag en skrivebordstest af programmet, hvor relevante registre og lagerceller vises efter udførslen af hver enkelt instruktion.

c)
Foretag en aftestning af programmet med brug af CODEVIEW, idet der køres "single-step", så der kan sammenlignes med skrivebordstest.


Programmet kan udleveres på diskette og hedder ASMPROC1

Opgave 11.

Udbyg det udleverede program med en rutine, der kan kon​vertere den binære sum til ASCII format, samt en rutine der kan udskrive den konverterede sum på skær​men.

DATA    Segment Para Public 'DATA'
Antal   Dw      5                       ; Antal el. i ITab

ITab    Dw      11,22,33,44,45

Sum     Dw      ?

DATA    EndS                            ; slut på data‑segment
CODE    Segment Para Public 'CODE'

        Assume Cs:CODE,Ds:DATA,Es:nothing,Ss:STACK

; Proceduren Main er den rutine der får kontrollen ved start

Main    Proc    Near
        Call    InitDS                  ; udfør InitDS 

        Call    Beregn                  ; udfør Beregn 

        Call    StopPG                  ; udfør StopPG

Main    EndP
; Proceduren Opstart opsætter adresse på datasegmentet

InitDS  Proc    Near
        Mov     Ax,Seg DATA             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Ret                             ; retur fra procedure

InitDS  EndP
; Proceduren Afslut overgiver kontrollen til DOS igen

StopPG  Proc    Near
        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

;  Da denne funktion ikke giver kontrollen tilbage

;  er der ingen Ret‑instruktion

StopPG  EndP
; Proceduren Beregn udregner summen af tal i ITab   

Beregn  Proc    Near
        Xor     Ax,Ax                   ; Ax := 0

        Xor     Si,Si                   ; Si ‑> første elm.

        Mov     Cx,Antal                ; FOR Cx := Antal

        Cmp     Cx,0                    ;  DOWNTO 1

For:    Jle     EndFor                  ;   2. runde Dec Cx

        Add     Ax,ITab[Si]             ;   Ax := Ax+ITab.elm.

        Add     Si,2                    ;   Si ‑> næste elm.

        Dec     Cx                      ; næste cx i for‑løkke

        Jmp     For

EndFor:                                 ; END

        Mov     Sum,Ax                  ; Sum := Ax

        Ret                             ; retur fra procedure

Beregn  EndP
CODE    EndS                            ; slut på code‑segment
STACK   Segment Para Stack 'STACK'
        dw      128 dup(?)              ; 128 ord afsat

STACK   EndS                            ; slut på stack‑seg
        End     Main                    ; Slut og opsæt start
Opgave 12.

Upper   Proc    Near


        Push    Ax              ; save arbejdsregister


        Push    Bx              ; save arbejdsregister


        Mov     Bx,Dx           ; Bx := adr. index


Loop1:


        Mov     Al,[Bx]         ; While "bogstav"


        Cmp     Al,'$'          ;    Ds:[Bx] <> '$'


        Je      EndLoop1        ;       Do


;       ......


;       ......


;       ......


        Inc     Bx              ;  Ds:Bx => næste bog.


        Jmp     Loop1           ; EndWhile


EndLoop1:


        Pop     Bx              ; restore register


        Pop     Ax              ; restore register


        Ret                     ; returner


Upper   EndP


Ovenstående er udgangspunkt for en rutine, der skal kunne rette alle bogstaver i en tekst til store bogstaver.


Ved kald af rutinen indeholder Ds:Dx adressen på første tegn i teksten. Teksten afsluttes med et $-tegn.


Du bedes gøre rutinen færdig.


Rutinen kan aftestes ved skrivebordstest, eller den kan aftestes maskinelt. Der kan udleveres et program (ASMUP​PER), der indeholder den påbegyndte rutine samt testkald.

Opgave 13.

Givet er følgende programstump:


0


1
        Mov     Bx,6


2
        Mov     Ax,1


3
L1:


4
        Mov     DataA[Bx],Al


5
        Shl     Ax,1           ; bits 1 plads left


6
        Dec     Bx


7
        Jnl     L1


8


DataA er defineret på følgende måde:


DataA   Db      16 Dup (0)


Skrivebordstest programstumpen.


Hvad indeholder DataA og de berørte registre (Ax og Bx) efter rutinen er kørt.

Opgave 14.
Givet er følgende definitioner i datasegmentet:


A       Db      '00123'


B       Db      '45678'


C       Db      6 dup(?)

Samt følgende programstump:


Linie        Programkode


01
P1:


02
        Mov     Bx,4


03
        Mov     Al,0


04
        Mov     Dl,10


05
        Mov     Dh,(48*2)+10   ; udregnes af


06
                               ; oversætter


07
        Mov     Cl,8


08
L1:


09
        Shl     Ax,Cl


10
        Add     Ah,A[Bx]


11
        Add     Ah,B[Bx]


12
        Sub     Ah,Dh


13
        Jl      L2


14
        Inc     Al


15
        Jmp     L3


16
L2:


17
        Add     Ah,Dl


18
L3:


19
        Or      Ah,48


20
        Mov     C+1[Bx],Ah


21
        Dec     Bx


22
        Jnl     L1


23



24
        Or      Al,48


25
        Mov     C,Al


26
        


27
        Ret

a)
Foretag en skrivebordstest af ovenstående programstump, idet alle relevante data (herunder registre) vises efter udførsel af de enkelte programlinier.

b)
Udarbejd en beskrivelse af ovenstående programstump.

Opgave 15.
Fremstil et assemblerprogram, der kan ændre farven på alle tegn på skærmen til grøn tekst på rød baggrund.

Programmet skal anvende direkte acces til skærmbufferen.

Skærmen kører i alm. tekstmode med 80 tegn pr. linie og 25 linier (2000 tegn).

Skærmbufferen er pladseret i adresse B800:0000

løsning i PASCAL.
Program NyATT;

VAR

    i,SkrmSeg,SkrmOfs: Integer;

BEGIN

     SkrmSeg := $B800;

     SkrmOfs := $0000;                      { b BAG i FOR } 

     FOR i := 1 to 2000 DO BEGIN            { l RGB n RGB }

          MEM[SkrmSeg:SkrmOfs+1] := $42;    { 0 100 0 010 }

          SkrmOfs := SkrmOfs + 2;

     END;

END.

Opgave 16.
Nedenstående viser en programstump i PASCAL til søgning i en usorteret tabel.


VAR



SoegeTal:
Integer;



AntalElm: Integer;



i:        Integer;



Fundet:   Boolean;



Tab:      Array[1..200] of Integer;


............


............


i := 0;


Fundet := False;


WHILE (NOT Fundet) AND (i < AntalElm) DO BEGIN



i := i + 1;



IF (SoegeTal = Tab[i]) THEN




Fundet := True;


END;


IF Fundet then



WRITELN('Fundet')


ELSE



WRITELN('Ikke Fundet');

Omskriv ovenstående programstump til 8086 assembler.

Når programstumpen kører er AntalElm sat til 200 og tabellen er blevet fyldt op.

Opgave 17.
Fremstil et assemblerprogram, der kan indlæse en dato fra tastaturet, beregne det tilhørende dagnr og udskrive dette på skærmen.

Opgave 18.
Fremstil et assemblerprogram, der kan hente dagsdato fra systemet med DOS-kald og udskrive månedens navn på skærmen.

(Januar, Februar ...... December)

Udbyg evt. programmet så også dag og årstal udskrives.

(eksempel. 10. Januar 1993)

Opgave 19.
Fremstil et assemblerprogram, der kan sammenligne første linie fra en "fil" (standard input) med en tekst, der angives som parameter til pro​grammet. Hvis der er forskel afleveres en returkode > 0 til DOS ellers retureres med 0.

Parameteren fås fra PROGRAM-SEGMENT-PREFIX'et (PSP), hvor der er afsat plads til en streng på 127 tegn. Længden af strengen er angivet i en byte lige før strengen (svarende til formatet i en STRING i PASCAL).

Ved opstart af såvel EXE- som COM-programmer vil Es og Ds indeholde segmentadres​sen på PSP'et.

Filen kan redirigeres eller pipes ind i programmet, således at dette kan læse fra standard input (som fra tastatureu).

DOS-funktion 0A kan anvendes såvel som 3F.

Den sidste funktion arbejder med et såkaldt HANDLE-nummer, der dels kan tildeles fra DOS når en fil åbnes (med en anden DOS-funktion 3D), dels knyttes til standard-enhederne:



Handle'nr:
Standardenhed




0

Standard input (keyboard, redir, pi​pe)




1

Standard output (Skærm, redir, pipe)




2

Standard error (skærm kan ikke under​trykkes med redir eller pipe)

Opgave 20.
Fremstil et program, der kan finde DOS'ens environment blok, hvor PROMPT, PATH m.m. findes og herefter udskrive indholdet af PATH parameteren.

Segmentadressen på environmentblokken kan findes via PSP'et, hvor adressen opbe​vares.

Opgave 21.
Fremstil et program (WRITEENV), der kan udskrive en parameter fra environ​ment​blokken.

Navnet på parameteren angives som parameter til programmet ved opstarten.

Eks.

WRITEENV PATH

Dette skulle give en udskrift af PATH-parameteren

Opgave 22.
Fremstil en assembler rutine, der kan rykke alle linier på skærmen en linie ned              (SC​ROLLE en linie baglæns).

Der skal anvendes BIOS-kald (INT 10)

Opgave 23.
Fremstil en assembler rutine, der kan slette en firkant midt på skærmen med øverste venstre hjørne i linie 10, kolonne 30 og nederste højre hjørne i linie 16, kolonne 60.

Der skal anvendes BIOS-kald (INT 10)

Opgave 24.
Fremstil en assembler rutine, der kan udskrive en tekst på skærmen ved brug af BIOS-kald.

Opgave 25.
Fremstil en assembler rutine, der kan indlæse et password fra tastaturen uden at det udskrives på skærmen.

Opgave 26.
Fremstil en assembler rutine, der kan teste om venstre eller højre shift-tast er trykket ned.

Opgave 27.
Fremstil et assembler program, der kan indlæse en ASCII-fil og udskrive en indholdet i en ny fil, idet alle bogstaver kon​verteres til store (anvend redirektion for input og output).

Opgave 28.
Fremstil en rutine, der kan udskrive en byteværdi binært (8 ascii 0/1-taller)

Opgave 29.
2-dimmensionel tabel:

           ------------ Xmax ----- --------        

           ┌────┬────┬────┬────┬ ────┬────┐        

    Y=1    │    │    │    │    │     │XSUM│   |    

           ├────┼────┼────┼────┼ ────┼────┤   |    

    Y=2    │    │    │    │    │     │XSUM│   |    

           ├────┼────┼────┼────┼ ────┼────┤        

    Y=3    │    │    │    │    │     │XSUM│   Ymax 

           ├────┼────┼────┼────┼ ────┼────┤        

    Y=n    │    │    │    │    │     │XSUM│   |    

           ├────┼────┼────┼────┼ ────┼────┤   |    

    Y=Ymax │YSUM│YSUM│YSUM│YSUM│ YSUM│YSUM│   |    

           └────┴────┴────┴────┴ ────┴────┘   |    

            X=1  X=2  X=3  X=4   X=n  X=Xmax       

Ovenstående viser en 2-dimensionel tabel af integer.

Tabellen er udlagt i lageret således at elementerne første række (Y) kommer efter hinanden, efter sidste felt i rækkken (sumfelt) følger første felt fra næste række.

Felterne i samme søjle (X) ligger således ikke ved siden af hinanden, men med en afstand svarende til længden af en række.

Xmax:
Angiver antal elementer i en række (antal søjler) incl. sumfelt.

Ymax:
Angiver antal elementer i en søjle (antal rækker) incl. sumfelt.

a)
Udarbejd en assemblerrutine, der kan beregne summen af hver række (XSUM) excl. sidste (Y=Ymax).


Løsningsskitse i PASCAL:



FOR Y := 1 to Ymax-1 DO BEGIN




SUM := 0;




FOR X := 1 to Xmax-1 DO BEGIN





SUM := SUM + TAB[Y,X];




END;




TAB[Y,X] := SUM;



END;

b)
Udarbejd en assemblerrutine, der kan beregne summen af hver søjle incl. den sidste (X=Xmax).


Løsningenskitse i Pascal ligner ovenstående temmelig meget, så derfor gives her nogle andre tips.


Da der skal springes tilbage i adresserne ved start på hver søjle, er det en ide, at fastholde adressen på før​ste elm. i den søjle der behandles. Starten på næste søjle kan nemt bereges da den er et elm. højere. Brug BP registeret til dette formål. 


Brug nu SI registret til at holde styr på den aktuelle element i søjlen, idet BP angiver start på søjle og SI afstand herfra. Af​standen mellem elementerne i søjlen, kan beregnes som XMAX*elm.læng​de.

Opgave 30.
Udarbejd et assemblerprogram PARCOUNT, der kan optælle antal parametre i parameterlisten, og returnere antallet som ERROR-CODE, der kan aftestes i DOS som IF ERRORLEVEL.

Opgave 31.
Udarbejd et assemblerprogram TRANSLATE, der kan anvendes som filter i DOS (læse std. input og skrive std. output).

Programmet skal omsætte alle bogstaver til store.

Opgave 32.
Udbyg foregående program, så der kan angives parameteren /L, hvorved alle bogstaver skal omsættes til små istedet for store. Hvis parameteren ikke er sat skal der stadig omsættes til store bogstaver.

Opgave 33.
Udbyg foregående program, så der kan angives parameteren /A, hvorved alle tegn mindre end blanke (32), med undtagelse af CR, LF og FF, samt alle tegn større end 127 skal konverteres til blanke.

Opgave 34.

01: P2      PROC    NEAR


02:         Mov     Dx,Cx


03:         Mov     Cl,4


04:         Xor     Ah,Ah


05:         Xor     Bh,Bh


06: P2L1:


07:         Xor     Di,Di


08:         Mov     Ch,8


09: P2L2:


10:         Mov     Al,[Si]


11:         Shl     Ax,Cl


12:         Shr     Al,Cl


13: R1:


14:         And     Ah,00001111B


15:         Mov     Bl,Ah


16:         Mov     Ah,DataC[Bx]


17:         Mov     DataB[Di],Ah


18: R2:


19:         Inc     Di


20:         Mov     Bl,Al


21:         Mov     Al,DataC[Bx]


22:         Mov     DataB[Di],Al


23:         Inc     Di


24:         Inc     Di


25:         Inc     Si


26:         Dec     Ch


27: R3:


28:         Jnz     P2L2


29:         Mov     Bx,Dx


30:         Mov     Dx,Offset DataB


31:         Mov     Ah,09h


32:         Int     21h


33:         Mov     Dx,Bx


34:         Cmp     Si,Dx


35:         Jl      P2L1


36:         Ret


37: P2      ENDP

Datasegment indeholder:

    DataA   Db      128 Dup (0)
; 128 byte med binær 0


    DataB   Db      '.. .. .. .. .. .. .. ..',13,10,'$'


    DataC   Db      '0123456789ABCDEF'

a)
Udfør en skrivebordstest af et gennemløb af assemblerru​tinen P2, når følgende oplyses:


-
DataA har fået nyt indhold, nemlig 4 ASCII-tegn "TEST" efterfulgt af 124 binære nuller.


-
Cx indeholder ved starten af P2 værdien 4.


-
Si indeholder offset på DataA, her 0.


Skrivebordstesten skal dokumenteres, ved at notere ind​holdet af registre og relevante lagerceller på følgende steder i programmet P2L1, P2L2, R1, R2, R3 samt ved af​slutningen af P2.

b)
Forklar hvad rutinen laver.

Opgave 35.
Udarbejd et assemblerprogram "LIXTAL" i 8086 macro-assembler.

Programmet skal kunne optælle antallet af ord og bogstaver/tal i en asciitekst. Desuden udreges gennemsnitsordlængden som antal bogstaver divideret med antal ord.

Der er tale om et meget forsimplet lix-tal.

Programmet skal udskrive de tre nøgletal med passende lede​tekst på skærmen.

Karakterer definers her som bogstaverne a-z,æ,ø,å,A-Z,Æ,Ø,Å samt tallene 0-9.

Et ord defineres som sammenhængende bogstaver/tal. Tegn der ikke er bogstaver/tal vil således adskille ord.

Teksten indlæses fra standard input, der kan være tastaturet eller en redirigeret fil.

Der kan indlæses med dos-funktion 0Ah, hvor der indlæses en linie af gangen, idet der så skal optræde en stopværdi til sidst (f.eks en tom linie eller et specialtegn).

Der kan også indlæses med dos-funktion 3Fh, hvor bufferen sættes stor nok til at kunne rumme hele teksten, der så kan indlæses på een gang.

Opgave 36.
Udarbejd et assemblerprogram "OPTAEL" i 8086 macro-assembler, som kan foretage en optælling af forekomsten af de enkelte ASCII-tegn (extended) i en tekstfil, der overføres som stan​dard-input evt. pipe eller redirektion.

Eksempel BREV.TXT som input:


OPTAEL < BREV.TXT

Optællingen skal kun omfatte tegnene i intervallet 20 (hex) til AF (hex).

Resultalet af optællingen skal udskrives på skærmen.

Tips !   Tips !   Tips !   Tips !
Opbyg programmet i mindst 4 procedurer:


1.
Styring.


2.
Indlæsning i buffer.


3.
Optælling for indlæst buffer.


4.
Udskrift af slutresultat.

Vedrørende indlæsning fra DOS standardinput henvises til dos interrupt beskrivelser.

Indexer et tegn af gangen fra input-buffer (f.eks med SI registeret).

Definer evt. inputbuffer, så den kan rumme helle filen (sæt max til 4096).

Man kan evt. istedet indlæse eet tegn af gangen og stoppe når dette er CTRL-Z (ascii tegn nr. 26).

Optælling kan foretages på to måder:
1.
Den pladsforbrugende men hurtige.


Der oprettes en tabel på 256 ord initieret med nuller.


Brug det indlæste tegn direkte som index til op​tællings​tabel ved at gange værdien med 2 (længden af et element) (brug shift instruktion), tæl ta​bel-element som nu kan indexeres op med 1.


Husk at ascii-tegnet er på 1 byte og index er på 2 byte, så vælg index-register med omhu.

2.
Den pladsbesparende men langsommere.


Der oprettes en tabel på 144 ord initieret med nuller (svarende til intervallet fra 20 hex til AF hex).


Der subtraheres 20 (hex) fra det indlæste tegn, hvis resultatet er negativt hoppes uden om optæl​ling.


Der testes herefter om tegn er > 144 (AF-20 hex), hvis dette er tilfældet hoppes ligeledes uden om optælling.


Den nye værdi anvendes herefter som index, idet der som før først ganges med 2.


Fortsættes næste side.
Udskriften kan foretages på følgende måde:

Udskriv en linie for hvert af de aktuelle ascii-tegn (et af gangen).


For hvert element i optællingstabellen foretages en konvertering fra binært til ascii. Der henvises til opgave med konvertering fra binær til ascii.


Det enkelste er nok at optælle ascii-tegn og index til tabel hver for sig. Ascii-tegnen kan gemmes sammen med den "stren​g" der anvendes til konvertering, således at de kan udskrives på en gang.


Husk, hvis optællingsmodel 1 anvendt, så skal index starte 20 (hex) * 2 inde i tabellen. Der kan dog også angives et øget offset på tabellen. 

Programmet kan evt. udbygges, så der ikke udskrives noget for uanvendte ascii-tegn.

For de meget entusiastiske kan programmet udbygges, således at der foretages en optælling af ordlængder (regn med max 20 tegn).

Et ord kan her defineres som afsluttet med en blank, et komma, et punktum samt alle de ascii-tegn, der ikke medtages i op​tællingen. Et ord starter med ethvert af de øvrige ascii-tegn.

Ved udbygning så start under alle omstændigheder med mindstekravet og udbyg så evt. senere.

Løsninger:   (Sidenumre passer ikke ved ændringer)
Courier 10p til resten

5.2.a


8086 assemblerprogrammer


Opgaveløsninger

Bjørk Busch
Opgave 1 løsning:
CODE     Segment

         Assume    Cs:CODE,Ds:DATA,Ss:STACK

Start:

         Mov       Ax,Seg DATA

         Mov       Ds,Ax                ; Etabler datasegnemt

         Mov       Ax,Tal1

         Add       Ax,TAl2              ;  Ax := (Tal1+Tal2)

         Mov       Bx,Ax                ;  Gem mellemresult

         Add       Ax,Bx                ;  ! * 2

         Add       Ax,Bx                ;  ! * 3

         Mov       Result,Ax            ;  store resultat

Slut:

         Mov       Al,0

         Mov       Ax,4Ch

         Int       21h                  ;  Returner til DOS

CODE     EndS                           ;  Slut på CODE‑segment

DATA     Segment

Tal1     Dw        32

TAl2     Dw        13

Result   Dw        ?

DATA     EndS

STACK    Segment STACK 'STACK'

         Dw        128 DUP (?)

STACK    EndS

         END       START                ;  Slut på program

                                        ;  Start i label START

Opgave 2 løsning:
CODE     Segment

         Assume    Cs:CODE,Ds:DATA,Ss:STACK

Start:

         Mov       Ax,Seg DATA

         Mov       Ds,Ax                ; Etabler datasegnemt

         Mov       Ax,Tal1

         Add       Ax,TAl2              ;  Ax := (Tal1+Tal2)

         Mov       Bx,Ax                ;  Gem mellemresult

         Add       Ax,Bx                ;  ! * 2

         Add       Ax,Bx                ;  ! * 3

;         Mov       Result,Ax            ;  store resultat

;         Mov       Ax,Result            ;  if

;         Cmp       Ax,0                 ;    Result <,=,> 0

          Jnl       TElseIf1

;         Mov       Ax,Result

          Add       Ax,100

          Mov       Result,Ax

          Jmp       TEndIf

TElseIf1:

;         Mov       Ax,Result

;         Cmp       Ax,0

          Jne       TElseIf2

          Mov       Ax,20

          Mov       Result,Ax

          Jmp       TEndIf

TElseIf2:

;          Mov       Ax,Result

          Sub       Ax,100

          Mov       Result,Ax

TEndIf:

Slut:

         Mov       Al,0

         Mov       Ax,4Ch

         Int       21h                  ;  Returner til DOS

CODE     EndS                           ;  Slut på CODE‑segment

DATA     Segment

Tal1     Dw        32

TAl2     Dw        13

Result   Dw        ?

DATA     EndS

STACK    Segment STACK 'STACK'

         Dw        128 DUP (?)

STACK    EndS

         END       START                ;  Slut på program

                                        ;  Start i label START

Opgave 3 løsning:
CODE     Segment

         Assume    Cs:CODE,Ds:DATA,Ss:STACK

Start:

         Mov       Ax,Seg DATA

         Mov       Ds,Ax                ; Etabler datasegnemt

         Mov       Ax,0                 ; SUM := 0

         Mov       Si,0                 ; Si -> første tabel-elm.

         Mov       Cx,4                 ; Loop-counter

Igen:

         Add       Ax,Itab[Si]          ; SUM := SUM + ITAB[i]

         Add       Si,2                 ; Si -> næste elm (2 byte pr. elm.)

         Sub       Cx,1

         Jg        Igen                 ; Forfra hvis flere gennemløb

         Mov       Sum,Ax               ; Gem SUM

Slut:

         Mov       Al,0

         Mov       Ax,4Ch

         Int       21h                  ;  Returner til DOS

CODE     EndS                           ;  Slut på CODE‑segment

DATA     Segment

ITAB     DW        11,21,31,41

SUM      DW        ?

DATA     EndS

STACK    Segment STACK 'STACK'

         Dw        128 DUP (?)

STACK    EndS

         END       START                ;  Slut på program

                                        ;  Start i label START

Der anvendes her en ikke mekanisk oversættelse, idet løkkestyring og sekventiel indexering i tabellen adskilles.

Det er ikke nødvendigt at teste loop-counter = 0, idet SUB sætter flag.

Istedet for SUB kunne man her anvende instruktionen DEC.
         Mov       SUM,0                 ; SUM := 0

For1Init:

         Mov       I,1

         Dec       I                    ; Juster for første test

For1Tst:

         Inc       I

         Cmp       I,4

         Jg        For1End

For1Beg:

         Mov       Si,I                 ; udregn index til offset-adresse

         Dec       Si                   ;      juster til start i 0 og

         Add       Si,Si                ;      multiplicer 2, der er

                                        ;      elementlængden

         Mov       Ax,Sum               ; SUM :=

         Add       Ax,Itab[Si]          ;       SUM + 

         Mov       Sum,Ax               ;             ITAB[i]

         Jmp       For1Tst

For1End:

Opgave 4+5+6 løsning:
Algoritme i Pascal:

Var

         Inavn: ARRAY[1..11] OF CHAR;

         Unavn: ARRAY[1..30] OF CHAR;

         ix:    INTEGER;

         ux:    INTEGER;

...................

         Inavn := 'EDB-SKOLEN$';

         Unavn := '                              ';

...................

         ix := 1;

         ux := 1;

         WHILE Inavn[ix] <> '$' do begin

              Unavn[ux] := Inavn[ix];

              ix := ix + 1;

              ux := ux + 2;

         END;

         WRITELN(Unavn);

Assemblerløsning:

CODE     Segment

         Assume    Cs:CODE,Ds:DATA,Ss:STACK

Start:

         Mov       Ax,Seg DATA

         Mov       Ds,Ax                ; Etabler datasegnemt

         Mov       Si,0                 ; Si -> første tegn i Inavn

         Mov       Di,0                 ; Di -> første tegn i Unavn

WhileTst:

         Mov       Al,Inavn[Si]

         Cmp       Al,'$'

         Je        EndWhile             ; Slut hvis $-tegn

WhileBeg:

         Mov       Unavn[Di],Al         ; Unavn[Di] := Inavn[Si]

         Add       Si,1

         Add       Di,2

         Jmp       WhileTst

EndWhile:

         Mov       Al,09h

         Lea       Dx,Unavn

         Int       21h

Slut:

         Mov       Al,0

         Mov       Ax,4Ch

         Int       21h                  ;  Returner til DOS

CODE     EndS                           ;  Slut på CODE‑segment

DATA     Segment

INAVN    DB        'EDB-SKOLEN','$'

UNAVN    DB        30 DUP (' ')

         DB        13,10,'$'

DATA     EndS

STACK    Segment STACK 'STACK'

         Dw        128 DUP (?)

STACK    EndS

         END       START                ;  Slut på program

                                        ;  Start i label START

Opgave 7 løsning:
Algoritme i Pascal:
VAR

     Tekst: ARRAY[1..?] OF CHAR;

     Antal: INTEGER;

     i:     INTEGER;

     ...................................

     ...................................

     Antal := 0;

     i := 1;

     While (Tekst[i] <> #0) DO BEGIN

          IF (Tekst[i] = 'A' THEN BEGIN

               Antal := Antal + 1;

          END;

          i := i + 1;

     END;

Assemblerløsning:
Tekst    Db        'Tekst med et antal a'er der optælles',0

AntalA   Dw        ?

         Mov       Ax,0

         Mov       Si,0               ; Si -> på første tegn 

While1Tst:

         Mov       Bl,Tekst[Si]       ; Hent udvalgt tegn

         Cmp       Bl,0

         Je        While1End

While1Beg:

If1Tst:

         Cmp       Bl,'A'             ; hvis tegn = 'A'

         Jne       If1End

If1Beg:

         Inc       Ax                 ; Adder 1 til antal A'er

If1End:

         Inc       Si                 ; Si -> Næste tegn  

         Jmp       While1Beg

While1End:

         Mov       AntalA,Ax          ; Gem antal A'er

Opgave 8 løsning:
        PAGE 72,120

; Programmør: Bjørk Busch       Dato: 28/3‑1993           

; Proceduren BIN2ASC konverterer et binært (word) til Ascii‑format

; Parametre:    Ax          binært Word der skal konv.

;               Cx          Antal cifre i ascii‑tal

;               Ds:Di       Startadresse på ascci‑tal (resultat)

BIN2ASC Proc    Near

        Push    Ax                      ; Gem registre, der

        Push    Bx                      ; ændres i proc.

        Push    Cx

        Push    Dx

        Push    Di

        Mov     Bx,10                   ; Grundtal i 10‑tal‑sys

        Add     Di,Cx                   ; Ds:Di := adr. på

        Dec     Di                      ;    sidste ciffer.

BinLoop:

        Xor     Dx,Dx                   ; nulstil Dx så

                                        ;     Dx/Ax = Ax

        Div     Bx                      ; Ax := Dx/Ax DIV 10

                                        ;    Ax kvotient

                                        ; Dx := Dx/Ax MOD 10

                                        ;    Dx rest

                                        ; Da 0<=rest<=9 kan man

                                        ;   regne med Dl alene

        Add     Dl,'0'                  ; Omregn ciffer til ascii

        Mov     [Di],Dl                 ;   flyt cif. til resultat

        Dec     Di                      ; Ds:Di ‑> nyt ascii‑cif.

        Loop    BinLoop                 ; Fortsæt hvis flere cifre.

        Pop     Di                      ; genetabler indhold

        Pop     Dx                      ; var ved rutinens start

        Pop     Cx                       

        Pop     Bx

        Pop     Ax

        Ret                             ; retur fra procedure

BIN2ASC EndP

Opgave 9 løsning:
        PAGE 72,120

; Programmør: Bjørk Busch       Dato: 28/8‑1993           

; Proceduren ASC2BIN konverterer et asciital til bibært (word)

; Parametre:    Ds:Si       Startadr. på ascii‑tal der skal konv.

;               Cx          Antal cifre i ascii‑tal

;               Ax          Resultat

ASC2BIN Proc    Near

        Push    Bx                      ; Gem registre, der

        Push    Cx                      ; ændres i proc.

        Push    Dx

        Push    Si

        Xor     Ax,Ax                   ; Ax := 0

        Mov     Bx,10                   ; Grundtal i 10‑tal‑sys

AscLoop:

        Mul     Bx                      ; Dx/Ax := Ax * 10

                                        ; result <= 65535 gør

                                        ; at Dx bliver 0 og 

                                        ; vi kan nøjes med 

                                        ; resultat i Ax

        Mov     Dl,[Si]                 ; Dl := næste asciitegn

        Sub     Dl,'0'                  ; Tegn omregnes til bin

        Add     Ax,Dx                   ; Ax := Ax + nyt ciffer

                                        ; NB! Dx og hermed Dh

                                        ; blev 0 i forb. med MUL

        Inc     Si                      ; Ds:Si ‑> nyt ascii‑cif.

        Loop    AscLoop                 ; Fortsæt hvis flere cifre.

        Pop     Si                      ; genetabler indhold

        Pop     Dx                      ; af registre, som det

        Pop     Cx                      ; var ved rutinens start

        Pop     Bx

        Ret                             ; retur fra procedure

ASC2BIN EndP

Opgave 12 program for udlevering:
        PAGE 66,120

        TITLE AsmUpper/opgave

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Programmør: Bjørk Busch       Dato: 20/10‑1992

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    Segment Para Public 'DATA'

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Data definitioner m.m.

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

CR      Equ     13                      ; carrige return

NL      Equ     10                      ; new line

Tekst   Db      'Denne tekst skal gerne omsættes til store '

        Db      ' bogstaver'

        Db      CR,NL,'$'

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    EndS                            ; slut på data‑segment

Cseg    Segment Para Public 'CODE'

        Assume Cs:Cseg,Ds:Dseg,Es:nothing,Ss:Sseg

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Hovedrutine

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Main    Proc    Near

        Mov     Ax,Seg Dseg             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Mov     Dx,Offset Tekst         ; DS:DX ‑> tekst

        Call    Skriv                   ; udfør procedure udskriv

        Call    Upper                   ; udfør procedure upper

        Call    Skriv                   ; udfør procedure udskriv

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4CH                  ; opsæt funktion for

        Int     21H                     ; terminate og udfør (MsDos)

Main    EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Slut på Hovedrutine

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Upper   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren omdanner små bogstaver til store i den tekst Ds:Dx peger

; på, idet teksten slutter med $

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push    Ax                      ; save arbejdsregister

        Push    Bx                      ; save arbejdsregister

        Mov     Bx,Dx                   ; Flyt adr. Dx til muligt idx

Loop1:

        Mov     Al,[Bx]                 ; While

        Cmp     Al,'$'                  ;    "bogstav" Ds:[Bx] <> '$'

        Je      EndLoop1                ;       Do

;       ......

;       ......

;       ......

        Inc     Bx                      ;  Ds:Bx => næste bogstav

        Jmp     Loop1                   ;  EndWhile

EndLoop1:

        Pop     Bx                      ; restore register

        Pop     Ax                      ; restore register

        Ret                             ; returner

Upper   EndP

Skriv   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren udskriver den tekst Ds:Dx peger på

; Rutinen anvender følgende registre som arbejdsregistre

;       Ax      som gemmes og restores ved afslutning

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push    Ax                      ; save arbejdsregister

        Mov     Ah,9H                   ; opsæt funktion for udskrift

        Int     21H                     ; af streng og udfør (MsDos)

        Pop     Ax                      ; restore register

        Ret                             ; returner

Skriv   EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Cseg    EndS                            ; slut på code‑segment

Sseg    Segment Para Stack 'STACK'

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Stack

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        dw      256 dup(?)              ; 256 ord sat af til stack

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Sseg    EndS                            ; slut på stack‑segment

        End     Main                    ; Slut og opsæt startadresse
Opgave 12 løsning - der indlægges følgende:
Algoritme i Pascal:
     IF (Tegn >= 'a') AND (Tegn <= 'z') THEN BEGIN

          Tegn := CHR( ORD(Tegn) - ORD('0') );

     END

     ELSE IF Tegn = 'æ' THEN BEGIN

          Tegn := 'Æ';

     END

     ELSE IF Tegn = 'ø' THEN BEGIN

          Tegn := 'Ø';

     END

     ELSE IF Tegn = 'å' THEN BEGIN

          Tegn := 'Å';

     END;

Assemblerløsning:
If1tst:

        Cmp     Al,'a'

        Jb      else1                   ; test uden fortegn

        Cmp     Al,'z'

        ja      else1

If1beg:

        Sub     Al,32                   ; ascii stor = ascii lille - 32

        Mov     [Bx],Al

        Jmp     If1end

Else1:

If2tst: Cmp     Al,'æ'

        Jne     Else2

If2beg:

        Mov     [Bx],'Æ'

        Jmp     If2end

Else2:

If3tst: Cmp     Al,'ø'

        Jne     Else3

If3beg:

        Mov     [Bx],'Ø'

        Jmp     If3end

Else3:

If4tst: Cmp     Al,'å'

        Jne     Else4

If4beg:

        Mov     [Bx],'Å'

        Jmp     If4end

if4end:

if3end;

if2end;

if1end;

Opgave 13 udleveret program:
        PAGE 66,120                                                

        TITLE AsmEks5  ‑  Opgave i skrivebordstest

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Programmør: Bjørk Busch       Dato: 22/3‑1992           

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    Segment Para Public 'DATA'                              

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

DataA   Db      16 Dup (0)

DataB   Db      32 Dup (255)

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    EndS                            ; slut på data‑segment  

Cseg    Segment Para Public 'CODE'                              

        Assume Cs:Cseg,Ds:Dseg,Es:Nothing,Ss:Sseg               

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Hovedrutine                                             

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Main    Proc    Near

        Mov     Ax,Seg Dseg             ; etabler               

        Mov     Ds,Ax                   ;     datasegment       

        Call    P1                      ; udfør procedure P1

        Call    P2                      ; udfør procedure P2

        Mov     Al,0                    ; returkode = 0         

        Mov     Ah,4CH                  ; opsæt funktion for    

        Int     21H                     ; terminate og udfør (MsDos)

Main    EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

P1      Proc    Near

        Mov     Bx,6

        Mov     Ax,1

L1:

        Mov     DataA[Bx],Al

        Shl     Ax,1                    ; Flyt bits 1 plads left

        Dec     Bx

        Jnl     L1

        Ret                             ; returner

P1      EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

P2      Proc    Near

        Mov     Di,Offset DataB

        Xor     Bx,Bx

L2:

        Mov     Ah,DataA[Bx]

        Mov     Al,bl

        Mov     Ds:[Di+Bx],Ax

        Inc     Di

        Inc     Bx

        Cmp     Ah,0

        Jnz     L2

        Ret                             ; returner

P2      EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Cseg    EndS                            ; slut på code‑segment  

Sseg    Segment Para Stack 'STACK'                              

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Stack                                                        

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        dw      256 dup(?)              ; 256 ord sat af til stack

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Sseg    EndS                            ; slut på stack‑segment 

        End     Main                    ; Slut og opsæt startadresse

Opgave 13 løsning.

Kun ændringer vises (efter programlinie er udført).


Flag vises kun på sidste instruktion, der retter før en test.


Linie    Ax     Bx     Flag    DATAA 16 decimale tal

     (PC)    dec.   dec.                                 


 0        ?      ?      ?      0   0   0   0   0   0   0 ..... 0

      1               6


 2        1

      3+4                                                   1


 5        2


 6               5      >0


 7


 3+4                                               2


 5        4


 6               4      >0


 7


 3+4                                           4


 5        8


 6               3      >0


 7


 3+4                                       8


 5       16


 6               2      >0


 7


 3+4                                   16


 5       32


 6               1      >0


 7


 3+4                              32


 5       64


 6               0      =0


 7


 3+4                          64


 5      128


 6              -1      <0


 7


 8


Hermed fås følgende slutværdier:


Ax:
128


Bx:
-1


DataA
:
64 32 16 8 4 2 1 0 0 0 0 0 0 0 0 0

Opgave 14 udleveret program:
        PAGE 66,120                                     

        TITLE AsmEks6  ‑  Debug øvelse / Skrivebordstest

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Programmør: Bjørk Busch       Dato: 27/3‑1992   

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    Segment Para Public 'DATA'                      

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

A       Db      '00123'

        Db      11 dup(0)

B       Db      '45678'

        Db      11 dup(0)

C       Db      6 dup(?)

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    EndS                    ; slut på data‑segment  

Cseg    Segment Para Public 'CODE'                      

        Assume Cs:Cseg,Ds:Dseg,Es:Nothing,Ss:Sseg       

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Hovedrutine                                     

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Main    Proc    Near

        Mov     Ax,Seg Dseg         ; etabler           

        Mov     Ds,Ax               ;     datasegment   

        Call    P1                  ; udfør procedure P1

        Mov     Al,0                ; returkode = 0     

        Mov     Ah,4CH              ; opsæt funktion for

        Int     21H                 ; terminate         

Main    EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

P1      Proc    Near

        Mov     Bx,4

        Mov     Al,0

        Mov     Dl,10

        Mov     Dh,(48*2)+10    ; udregnes af oversætter

        Mov     Cl,8

L1:

        Shl     Ax,Cl

        Add     Ah,A[Bx]

        Add     Ah,B[Bx]

        Sub     Ah,Dh

        Jl      L2

        Inc     Al

        Jmp     L3

L2:

        Add     Ah,Dl

L3:

        Or      Ah,48

        Mov     C+1[Bx],Ah

        Dec     Bx

        Jnl     L1

        Or      Al,48

        Mov     C,Al

        Ret

                                     ; returner

P1      EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Cseg    EndS                    ; slut på code‑segment  

Sseg    Segment Para Stack 'STACK'                      

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Stack                                           

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        dw      256 dup(?)    ; 256 ord sat af til stack

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Sseg    EndS                  ; slut på stack‑segment 

        End     Main        ; Slut og opsæt startadresse
Opgave 15 Løsning:
        PAGE 72,120

        TITLE  NYATT

;=========================================================

;  Retter attribut direkte i skærmbuffer

;=========================================================

CODE    Segment Para Public 'CODE'

;=========================================================

        Assume Cs:CODE,Ds:nothing,Es:nothing,Ss:STACK

Main    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,0B800h

        Mov     Es,Ax

        Mov     Di,0

        Mov     Cx,2000

        Mov     Ah,42h

For1:   Mov     Es:[Di+1],Ah

        Add     Di,2

        Loop    For1

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

;=========================================================

CODE    EndS                       ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;=========================================================

        dw      256 dup(?)         ; der er sat 128 ord af

;=========================================================

STACK   EndS                       ; slut på stack‑segment

        End     Main               ; Slut og opsæt startadr.

Opgave 16 Løsning:
Tab       Dw       200 DUP (?)

AntalElm  Dw       200

SoegeTal  Dw       ?


............


............

        Mov      Cx,0                 ; opsæt tæller og

        Mov      Si,-2                ;    tilhørende offset-index

        Mov      Bx,0                 ; Fundet := False

While1Tst:

        Cmp      Bx,0                 ;  Stop hvis

        Jne      While1End            ;   fundet

        Cmp      Cx,AntalElm          ;  eller

        Jnl      While1End            ;    ikke i < AntalElm

While1beg:

        Inc      Cx                   ; juster tæller og

        Add      Si,2                 ;   tilhørende offset-index

If1Tst:

        Mov      Ax,Soegetal

        Cmp      Ax,Tab[Si]

        Jne      If1End

If1beg:

        Mov      Bx,1                 ; Fundet := True

If1End:

        Jmp      While1Tst

While1End:

If2Tst:

        Cmp      Bx,1

        Jne      If2Else

If2Beg:

        Lea      Dx,Tekst1

        Mov      Ah,09h

        Int      21h

        Jmp      If2End

If2Else:

        Lea      Dx,Tekst2

        Mov      Ah,09h

        Int      21h

If2End:

        Jmp      If2End

Opgave 17 Løsning:

        TITLE  DAGNR

;===========================================================

;  Læser dato fra tastatur format DDMMÅÅ

;  Beregner og udskriver tilhørende dagnr

;===========================================================

DATA    Segment Para Public 'DATA'

;===========================================================

ReadBuf Label   BYTE

ReadMax Db      7

ReadLen Db      ?

ReadDat Db      7 Dup (?)

UdTxt   Db      13,10,'Dagnr='

UdDagnr Db      3 DUP (' ')

        Db      10,13,'$',100 DUP ('$')

MdTab   Dw      0                                  ; Jan

        Dw      0+31                               ; Feb

        Dw      0+31+28                            ; Mar

        Dw      0+31+28+31                         ; Apr

        Dw      0+31+28+31+30                      ; Maj

        Dw      0+31+28+31+30+31                   ; Jun

        Dw      0+31+28+31+30+31+30                ; Jul

        Dw      0+31+28+31+30+31+30+31             ; Aug

        Dw      0+31+28+31+30+31+30+31+31          ; Sep

        Dw      0+31+28+31+30+31+30+31+31+30       ; Okt

        Dw      0+31+28+31+30+31+30+31+31+30+31    ; Nov

        Dw      0+31+28+31+30+31+30+31+31+30+31+30 ; Dec

;=========================================================

DATA    EndS                            ; slut på data‑segment

CODE    Segment Para Public 'CODE'

;=========================================================

        Assume Cs:CODE,Ds:DATA,Es:nothing,Ss:STACK

INCLUDE ASC2BIN.ASM

INCLUDE BIN2ASC.ASM

Main    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,Seg DATA             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Call    Behandl

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

Behandl Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Egentlig behandling

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Call    ReadLN          ; Read DDMMÅÅ

        Cmp     ReadLen,6

        Jl      EndBehandl

        Lea     Si,ReadDat+2    ; Si ‑> MM

        Mov     Cx,2            ; 2 cifre

        Call    Asc2Bin         ; Ax := MM

        Mov     Bx,Ax           ; Bx := MM

        Dec     Bx              ; Juster til md‑start=0

        Mov     Dx,Bx           ; Gem MM

        Shl     Bx,1            ; Omregn til tabeladr.

        Mov     Bx,MdTab[Bx]    ; Ax := Dagnr for MD start       

        Lea     Si,ReadDat+0    ; Si ‑> DD

        Mov     Cx,2            ; 2 cifre

        Call    Asc2Bin         ; Ax := DD

        Add     Bx,Ax           ; Bx := Dagnr

        Lea     Si,ReadDat+4    ; Si ‑> ÅÅ

        Mov     Cx,2            ; 2 cifre

        Call    Asc2Bin         ; Ax := DD

        AND     Ax,00000011b    ; Test for skudår

        Jnz     TilAsc

        Mov     Ax,Dx           ; Justeret MM

        Dec     Ax

        Jz      TilAsc          ; > februar ?

        Inc     Bx              ; Skuddag

TilAsc:

        Mov     Ax,Bx

        Lea     Di,UdDagnr

        Mov     Cx,3

        Call    Bin2Asc

        Mov     Ah,09h

        Lea     Dx,UdTxt

        Int     21h        

EndBehandl:

        Ret

Behandl EndP

ReadLN  Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Indlæs linie fra STD input

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ah,0Ah             ; ReadLN

        Lea     Dx,ReadBuf

        Int     21h

        Ret

ReadLN  EndP

;=========================================================

CODE    EndS                       ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;=========================================================

        dw      256 dup(?)         ; der er sat 128 ord af

;=========================================================

STACK   EndS                       ; slut på stack‑segment

        End     Main               ; Slut og opsæt startadr.

Opgave 18 Løsning:

        TITLE  DAGSDATO

;===========================================================

;  Henter systemdato og udskriver

;===========================================================

DATA    Segment Para Public 'DATA'

;===========================================================

Mdlen   EQU     10

MdTab   Db      'Januar$   '

        Db      'Februar$  '

        Db      'Marts$    '

        Db      'April$    '

        Db      'Maj$      '

        Db      'Juni$     '

        Db      'Juli$     '

        Db      'August$   '

        Db      'September$'

        Db      'Oktober$  '

        Db      'November$ '

        Db      'December$ '

NyLin   Db      13,10,'$'       ; ANSI ESC‑sekvens

Dag     Db      '  .$'

AAr     Db      '    $'

Space   Db      ' $'

;===========================================================

DATA    EndS                            ; slut på data‑segment

CODE    Segment Para Public 'CODE'

;===========================================================

        Assume Cs:CODE,Ds:DATA,Es:nothing,Ss:STACK

INCLUDE BIN2ASC.ASM

Main    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,Seg DATA             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Call    Behandl

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

Behandl Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Egentlig behandling

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ah,2Ah          ; Get Date

        Int     21h

        Push    Cx              ; Gem år

        Push    Dx              ; Gem måned

        Mov     Al,Dl

        Mov     Ah,0            ; Ax := dag

        Lea     Di,Dag          ; Di ‑> ascii dag

        Mov     Cx,2

        Call    Bin2Asc         ; omsæt dag til ascii

        Lea     Dx,Dag

        Call    Write

        Lea     Dx,Space

        Call    Write

        Pop     Ax              ; hent måned i Bh

        Mov     Al,Ah

        Mov     Ah,0            ; Bx := måned

        Dec     Ax              ; omregn til start i 0

        Mov     Bx,Mdlen

        Mul     Bx              ; omregn til tabelindex

        Lea     Dx,Mdtab

        Add     Dx,Ax

        Call    Write

        Pop     Ax              ; hent årstal

        Lea     Di,AAr

        Mov     Cx,4

        Call    Bin2Asc

        Lea     Dx,Space

        Call    Write

        Lea     Dx,AAr

        Call    Write

        Lea     Dx,NyLin

        Call    Write

        Ret

Behandl EndP

Write   Proc    Near

        Push    Ax

        Mov     Ah,09h

        Int     21h        

        Pop     Ax

        Ret

Write   EndP

;===========================================================

CODE    EndS                       ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;===========================================================

        dw      256 dup(?)         ; der er sat 128 ord af

;===========================================================

STACK   EndS                       ; slut på stack‑segment

        End     Main               ; Slut og opsæt startadr.

Opgave 20,21 Løsning:

        TITLE WriteENV  ‑   udskrift af environment variabel

Sseg    Segment Para Stack 'STACK'

        dw      256 dup(?)              ; 256 ord sat af til stack

Sseg    EndS                            ; slut på stack‑segment

Dseg    Segment Para Public 'DATA'

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Data definitioner m.m.

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

CR        Equ     13                      ; carrige return

NL        Equ     10                      ; new line

Parm      Db      80 DUP ('$')

FejlTxt   Db      CR,NL,'Den angivne parameter findes ikke',CR,NL,'$'

FejlTxt2  Db      'Der ikke angivet nogen parameter !',CR,NL,'$'

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    EndS                            ; slut på data‑segment

Cseg    Segment Para Public 'CODE'

        Assume Cs:Cseg,Ds:Dseg,Es:nothing,Ss:Sseg

HentParm Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren henter variabelnavnet fra parameterlisten til

; test/udskrifts‑variablen PARM 

; zeroflag sat hvis der slet parameter var

; Arbejdsregistre sikres ikke

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Xor     Di,Di              ; tegnnr pointer

        Mov     Bp,0080h           ; start pointer parameterliste ‑ 1

        Mov     Cl,Es:[0080h]      ; hent længden af parameterlisten

        Inc     Cl                 ; +1 , da der startes med nedtælling

SkipBlanke:

        Dec     Cl                 ; 1 tegn mindre

        Jz      IngenParm

        Inc     Bp                 ; første / næste tegn

        Cmp     Byte Ptr Es:[Bp],' '

        Je      SkipBlanke

HentIgen:

        Dec     Cl                 ; 1 tegn mindre

        Jl      hentSlut

        Mov     Al,Es:[Bp+Di]      ; flyt fra parameterliste

        Mov     Parm[Di],Al        ;   til test/udskrifts variabel

        Inc     Di                 ; næste tegn

        Jmp     HentIgen

hentSlut:

        Mov     Parm[Di],'='       ; evt. 2 stk. = får ingen

                                   ;       betydning

IngenParm:

        Ret

HentParm EndP

TestParm Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren tester om variablen i PARM og i environment‑blokket har 

; samme navn                     

;  Input:       Es:[si]  start på parameter i environment‑blok

;  Output: OK =>    Al = '=' og zero‑flag sat,

;                   Es:[Si] og Parm[Di] udpeger begge '='

;          FEJL =>  Al = FF og zero‑flag ikke sat

;                   Es:[Si] og Parm[Di] udpeger første uoverens‑

;                   stemmelse

;  Arbejdsregistre sikres ikke

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Xor     Di,Di                   ; klargøring af pegepind

TestIgen:

        Mov     Al,Es:[Si]              ; Al := tegn fra ENV.blok

        Cmp     Al,Parm[Di]             ; Samme som i parm ??

        Jne     Uens                    ; Stemmer ikke

        Cmp     Al,'='                  ; Slut ??

        Je      Ens                     ; så var de ens

        Inc     Si                      ; næste tegn i ENV.blok

        Inc     Di                      ; næste tegn i Parm

        Jmp     TestIgen                ; forfra med næste tegn

Uens:

        Or      Al,0FFh                 ; sæt til uens (not zero)

Ens:

        Ret

TestParm Endp

NyParm  Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren flytter Si frem til næste parameter i environment‑

; blokken (første tegn)

; Hvis slut på blokken vil dette være et binært nul

; Arbejdsregistre sikres ikke

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Next:

        Cmp     Byte Ptr Es:[Si],0

        Je      NulFundet

        Inc     Si

        Jmp     Next

NulFundet:

        Inc     Si                      ; Flyt til tegn efter NUL

        Ret

NyParm  EndP

FlytParm Proc   Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren flytter parameteren fra environmentblokken til

; test/udskrifts variablen PARM.

; Ved start peger ES:[Si] og Parm[Di] på '='

; Klargører også PARM til udskrift.

; Arbejdsregistre sikres ikke

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Inc     Si

        Inc     Di

FlytLoop:

        Mov     Al,Es:[Si]              ; test

        Cmp     Al,0                    ;   om slut på ENV.parm

        Je      SlutFlyt

        Mov     parm[Di],Al             ; flyt til udskrift

        Inc     Si                      ; næste tegn i ENV.blok

        Inc     Di                      ; næste tegn i udskriftsvar.

        Jmp     FlytLoop

SlutFlyt:

        Mov     Parm[Di],CR             ; tilføj skift til

        Mov     Parm+1[Di],NL           ;  start på ny linie

        Ret

Flytparm EndP

UdSkriv Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren udskriver den tekst Ds:Dx peger på

; Rutinen anvender følgende registre som arbejdsregistre

;       Ax      som gemmes og restores ved afslutning

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push    Ax                      ; save arbejdsregister

        Mov     Ah,9H                   ; opsæt funktion for udskrift

        Int     21H                     ; af streng og udfør (MsDos)

        Pop     Ax                      ; restore register

        Ret                             ; returner

UdSkriv EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Hovedrutine

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Main    Proc    Near

        Mov     Ax,Seg Dseg             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Call    HentParm                 ; Hent variabelnavn fra

                                        ;    parameterlisten

        Jz      ParmFejl                ; der mangler parameter

        Mov     Es,Es:[002Ch]           ; Opsæt adresse til

                                        ; ENVIRONMENT blok

                                        ;   (via PSP offset 2Ch)

        XOR     Si,Si                   ;  Es:[Si] ‑> ENV.blok

Soeg:   Cmp     Byte Ptr Es:[Si],0      ; Test for stop på ENV.blok

        Je      IkkeFundet

        Call    TestParm                ; Test om det er den

        Je      Fundet

        Call    NyParm                  ; Find næste i ENV.blok

        Jmp     Soeg

Fundet:

        Call    Flytparm                ; Flyt fra ENV. til udskrift

        Lea     Dx,Parm

        Call    Udskriv

        Mov     Al,0                    ; returkode = 0 (ok)

        Jmp     Slut

IkkeFundet:

        Lea     Dx,Parm

        Call    Udskriv

        Lea     Dx,FejlTxt

        Call    Udskriv

        Mov     Al,1                    ; returkode = 1 (fejl)

        Jmp     Slut

parmfejl:

        Lea     Dx,FejlTxt2

        Call    Udskriv

        Mov     Al,2                    ; returkode = 2 (fejl)

Slut:

        Mov     Ah,4CH                  ; opsæt funktion for

        Int     21H                     ; terminate og udfør (MsDos)

Main    EndP

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Slut på Hovedrutine

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Cseg    EndS                            ; slut på code‑segment

        End     Main                    ; Slut og opsæt startadresse

Opgave 25 løsning:
Algoritme i Pascal:
     PassWord := '';

     i := 1;

     REPEAT

          Tegn := READKEY;

          IF Tegn <> #13 THEN BEGIN

               PassWord[i] := Tegn;

               Write('*');

               i := i + 1;

          END;

     Until Tegn = #13;

Assemblerløsning:
        Mov    Bx,0

Repeate:

        Mov    Ah,08h                ; læs tegn uden echo

        Int    21h

Iftst:

        Cmp    Al,13h

        Je     ifend;

Ifbeg:

        Mov    PassWord[Bx],Al

        Mov    Dl,'*'      

        Mov    Ah,02h                 ; skriv tegn i dl

        Int    21h

        Inc    Bx

Ifend:

until:  Cmp    Al,13

        Jne    Repeate

Opgave 28 Løsning:

        TITLE  BINDUMP

;===========================================================

;  procedure til binær dump og test af procedure

;===========================================================

DATA    Segment Para Public 'DATA'

;===========================================================

BinUd   Db      '        $'

NyLin   Db      13,10,'$'       ; ANSI ESC‑sekvens

Space   Db      ' $'

;===========================================================

DATA    EndS                            ; slut på data‑segment

CODE    Segment Para Public 'CODE'

;===========================================================

        Assume Cs:CODE,Ds:DATA,Es:nothing,Ss:STACK

INCLUDE BIN2ASC.ASM

Main    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,Seg DATA             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Mov     Al,01101100b            ; test

        Lea     Di,BinUd

        Call    DumpBin

        Lea     Dx,BinUd

        Call    Write

        Lea     Dx,NyLin

        Call    Write

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

DumpBin Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Dump binært          

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push    Ax

        Push    Di

        Mov     Cx,8

ForBeg:

        Mov     Ah,0            ;Ax := Al

        Shl     Ax,1

        Add     Ah,'0'          ; Ah := bitværdi i ascii

        Mov     [Di],Ah

        Inc     Di

        Loop    ForBeg

        Pop     Di

        Pop     Ax

        Ret

DumpBin EndP

Write   Proc    Near

        Push    Ax

        Mov     Ah,09h

        Int     21h        

        Pop     Ax

        Ret

Write   EndP

;===========================================================

CODE    EndS                       ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;===========================================================

        dw      256 dup(?)         ; der er sat 128 ord af

;===========================================================

STACK   EndS                       ; slut på stack‑segment

        End     Main               ; Slut og opsæt startadr.

Opgave 29 Løsning:

        TITLE  TAB2DIM

;===========================================================

;  Udregn række‑ og søjlesum i 2‑dimensionel tabel

;     &

;  Udskrift af 2‑dimensionel tabel

;===========================================================

DATA    Segment Para Public 'DATA'

;===========================================================

Xmax    Dw      6

Ymax    Dw      5

Tab     Dw      01,02,03,04,05,00

        Dw      02,04,06,08,10,00

        Dw      03,06,09,12,15,00

        Dw      04,08,16,20,24,00

        Dw      00,00,00,00,00,00

AntCif  Dw      4

AscTal  Db      '     $'

NyLin   Db      13,10,'$'

;===========================================================

DATA    EndS                            ; slut på data‑segment

CODE    Segment Para Public 'CODE'

;===========================================================

        Assume Cs:CODE,Ds:DATA,Es:nothing,Ss:STACK

INCLUDE BIN2ASC.ASM

Main    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,Seg DATA             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Call    XSum

        Call    YSum

        Call    Udskriv

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

XSum    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Beregning af rækkesum

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Dx,XMax

        Shl     Dx,1          ; Dx := Rækkelængde

        Mov     Bp,0          ; Rækkestart

        Mov     Cx,Ymax       ; Cx := antal rækkegennemløb

        Dec     Cx            ;     = Ymax ‑ 1

ForY1:

        Push    Cx            ; gem Y tæller

        Push    Dx            ; gem rækkelængde

        Mov     Ax,0          ; Sum := 0

        Mov     Si,0          ; Søjle index

        Mov     Cx,Xmax       ; Cx := antal søjlegennemløb

        Dec     Cx            ;     = Xmax ‑ 1

ForX1:

        Push    Cx            ; gem X tæller

        Add     Ax,Tab[Bp+Si] ; Ax := element

        Add     Si,2          ; Næste elm.

        Pop     Cx            ; restore X tæller

        Loop    ForX1

        Mov     Tab[Bp+Si],Ax ; gem rækkesum

        Pop     Dx            ; restore rækkelængde

        Add     Bp,Dx         ; Bp ‑> næste række

        Pop     Cx            ; restore Y tæller

        Loop    ForY1

        Ret

XSum EndP

YSum    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Beregning af søjlesum

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Dx,XMax

        Shl     Dx,1          ; Dx := Rækkelængde

        Mov     Bp,0          ; Søjlestart

        Mov     Cx,Xmax       ; Cx := antal søjlegennemløb

ForX2:

        Push    Cx            ; gem X tæller

        Mov     Ax,0          ; Sum := 0

        Mov     Si,0          ; række index

        Mov     Cx,Ymax       ; Cx := antal rækkegennemløb

        Dec     Cx            ;     = Ymax ‑ 1

ForY2:

        Push    Cx            ; gem Y tæller

        Add     Ax,Tab[Bp+Si] ; Ax := element

        Add     Si,Dx         ; Næste elm. i søjle

        Pop     Cx            ; restore X tæller

        Loop    ForY2

        Mov     Tab[Bp+Si],Ax ; gem søjlesum

        Add     Bp,2          ; Bp ‑> start på næste søjle

        Pop     Cx            ; restore Y tæller

        Loop    ForX2

        Ret

YSum EndP

Udskriv  Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Udskriv

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Dx,XMax

        Shl     Dx,1          ; Dx := Rækkelængde

        Mov     Bp,0          ; Rækkestart

        Mov     Cx,Ymax       ; Cx := antal gennemløb Ymax

ForY3:

        Push    Cx            ; gem Y tæller

        Mov     Si,0          ; Søjle index

        Mov     Cx,Xmax       ; Cx := antal gennemløb Xmax

ForX3:

        Push    Cx            ; gem X tæller

        Mov     Ax,Tab[Bp+Si] ; Ax := element

        Call    WriteTal

        Add     Si,2          ; Næste elm.

        Pop     Cx            ; restore X tæller

        Loop    ForX3

        Call    WriteLN

        Add     Bp,Dx         ; Bp ‑> næste række

        Pop     Cx            ; restore Y tæller

        Loop    ForY3

        Ret

Udskriv EndP

WriteTal Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Udskriv Tal i Ax

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push     Cx

        Push     Dx

        Push     Di

        Lea      Di,AscTal

        Mov      Cx,AntCif

        Call     Bin2Asc

        Lea      Dx,Asctal

        Call     Write

        Pop      Di

        Pop      Dx

        Pop      Cx

        Ret

WriteTal EndP

WriteLn Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Linieskift

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push     Dx

        Lea      Dx,NyLin

        Call     Write

        Pop      Dx

        Ret

WriteLn EndP

Write   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Udskriv under DOS

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push     Ax

        Mov      Ah,09h

        Int      21h

        Pop      Ax

        Ret

Write   EndP

;===========================================================

CODE    EndS                       ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;===========================================================

        dw      256 dup(?)         ; der er sat 128 ord af

;===========================================================

STACK   EndS                       ; slut på stack‑segment

        End     Main               ; Slut og opsæt startadr.

Opgave 30 Løsning:

        TITLE  COUNTPAR

CODE    Segment Para Public 'CODE'

;===========================================================

        Assume Cs:CODE,Ss:STACK

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Main    Proc    Near

        Call    Beregn

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

Beregn  Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Proceduren Beregn

; optæller antal parametre i parameterlisten (PSP)

; antal retur i Al

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Al,0

        Mov     Si,80h

        Mov     Cl,Es:80h   ;Cl := længde af parameterlisten

Skipblank:                              ; Spring over

        Inc     Si                      ; blanke

        Dec     Cl                      ; stop hvis

        Jl      Slut                    ; slut på

        Cmp     Byte Ptr Es:[Si],' '    ; parameter‑

        Je      SkipBlank               ; liste

Tael:   Inc     Al                      ; en parameter mere

SkipParm:

        Cmp     Byte Ptr Es:[Si],' '    ; Spring videre

        Je      SkipBlank               ; videre til

        Inc     Si                      ; slut på 

        Dec     Cl                      ; parameter

        Jl      Slut                    ; eller slut på

        Jmp     SkipParm                ; param. liste

Slut:                

        Ret                             ; retur fra procedure

Beregn  EndP

;===========================================================

CODE    EndS                            ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;===========================================================

        dw      128 dup(?)              ; der er sat 128 ord af

;===========================================================

STACK   EndS                            ; slut på stack‑segment

        End     Main                    ; Slut og opsæt startadr.

Opgave 31,32,33 Løsning:

        TITLE TRANSLATE

;===========================================================

;  TRANSLATE kan anvendes som filter

;  Uden parameter oversættes små bogstaver tol store (a‑z)

;  Med parameteren /L  omsættes store bogstaver til små

;  Med parameteren /A  accepteres kun CR,LF,FF og

;      ascii tegn fra 32‑127 ‑ øvrige omsættes til blanke

;===========================================================

DATA    Segment Para Public 'DATA'

;===========================================================

UpLo    Db      'U'                     ; U=upper, L=lower

AscTr   Db      ' '                     ; A=Ascii ‑ styretegn

Buffer  Db      512 DUP (?)

ErrTxt  Db      'Fejl ved læsning på STD‑input',13,10,'$'

;===========================================================

DATA    EndS                            ; slut på data‑segment

CODE    Segment Para Public 'CODE'

;===========================================================

        Assume Cs:CODE,Ds:DATA,Es:nothing,Ss:STACK

Main    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

; Main er den rutine der får kontrollen ved start

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,Seg DATA             ; etabler

        Mov     Ds,Ax                   ;     datasegment

        Call    GetParm

        Call    Behandl

        Mov     Al,0                    ; returkode = 0

        Mov     Ah,4Ch                  ; opsæt funktion for

        Int     21h                     ; terminate og udfør

Main    EndP

GetParm Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Opsæt styreparametre

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ax,'Ll'

        Call    CheckParm

        Cmp     Ax,0

        Jnz     NextParm

        Mov     UpLo,'L'

NextParm:

        Mov     Ax,'Aa'

        Call    CheckParm

        Cmp     Ax,0

        Jnz     EndGetParm

        Mov     AscTr,'A'

EndGetParm:

GetParm EndP

CheckParm Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Kontrol af parameter fra dos‑parameterliste   /p    

;                   hvor p er parameteroption

;  Input   Ah og Al    Bogstaver som skal valideres

;  Retur   Ax = 0 hvis parameter var i listen

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Si,80h

        Mov     Cl,Es:80h

FindParm:

        Inc     Si

        Dec     Cl

        Jl      Slut

        Cmp     Byte Ptr Es:[Si],'/'

        Jne     FindParm

Parameter:

        Cmp     Cl,1

        Jl      Slut

        Cmp     Es:[Si+1],Ah

        Je      Fundet

        Cmp     Es:[Si+1],Al

        Je      Fundet

        Jmp     FindParm

Fundet:

        Mov     Ax,0

Slut:                

        Ret                             ; retur fra procedure

CheckParm EndP

Behandl Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Egentlig behandling

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Call    Read                    ; Read first

While:

        Mov     Cx,Ax                   ; Antal tegn læst

        Jcxz    EndWhile

        Call    Convert

        Call    Write

        Call    Read                    ; Read next

        Jmp     While

EndWhile:

        Ret

Behandl EndP

Read    Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Indlæs fra STD input

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ah,3Fh              ; Read med handle

        Mov     Bx,0                ; Handle 0 = STD inp.

        Mov     Cx,512

        Lea     Dx,Buffer

        Int     21h

        Jnc     EndRead

        Mov     Ah,09h

        Lea     Dx,ErrTxt

        Int     21h

EndRead:

        Ret

Read    EndP

Write   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Skriv på STD output

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Mov     Ah,40h              ; Write med handle

        Mov     Bx,1                ; Handle 1 = STD outp.

        Lea     Dx,Buffer

        Int     21h

        Ret

Write   EndP

Convert Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  behandling af buffer  Cx = antal tegn  (Cx > 0)

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Push    Cx              ; gem antal

        Mov     Si,0            ; Si ‑> første tegn

ForBeg:

        Mov     Al,Buffer[Si]

TestLo:

        Cmp     UpLo,'L'

        Je      RetLo

        Call    Upper

        Jmp     TestAsc

RetLo:

        Call    Lower

TestAsc:

        Cmp     AscTr,'A'

        Jne     RetTegn

        Call    Ascii

RetTegn:

        Mov     Buffer[Si],Al

        Inc     Si              ; Si ‑> næste tegn

        Loop    ForBeg

        Pop     Cx

        Ret

Convert EndP

Upper   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Translate tegn i Al til uppercase

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Cmp     Al,'a'

        Jb      EndUpper

        Cmp     Al,'z'

        Ja      EndUpper

        Sub     Al,'a'‑'A'

EndUpper:

        Ret

Upper EndP

Lower   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;       Translate tegn i Al til lowercase

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Cmp     Al,'A'

        Jb      EndLower

        Cmp     Al,'Z'

        Ja      EndLower

        Add     Al,'a'‑'A'

EndLower:

        Ret

Lower EndP

Ascii   Proc    Near

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

;  Fjern styretegn (dog ikke LF,CR,FF) og tegn > 127

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        Cmp     Al,10

        Je      EndAscii

        Cmp     Al,13

        Je      EndAscii

        Cmp     Al,12

        Je      EndAscii

        Cmp     Al,' '

        Jb      MovSpace

        Cmp     Al,127

        Jbe     EndAscii

MovSpace:

        Mov     Al,' '

EndAscii:

        Ret

Ascii   EndP

;===========================================================

CODE    EndS                            ; slut på code‑segment

STACK   Segment Para Stack 'STACK'

;===========================================================

        dw      128 dup(?)              ; der er sat 128 ord af

;===========================================================

STACK   EndS                            ; slut på stack‑segment

        End     Main                    ; Slut og opsæt startadr.

Opgave 34 program for udlevering:
Cseg    Segment Para Public 'CODE'                            

        Assume Cs:Cseg,Ds:Dseg,Es:Nothing,Ss:Sseg             

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Main    Proc    Near             ; Start på hovedrutine       

        Mov     Ax,Seg Dseg      ; etabler                    

        Mov     Ds,Ax            ;     datasegment            

        Mov     Si,Offset DataA  ; Ds:Si = Adr. på DataA      

        Call    P1               ; Udføre P1                  

ML1:    Jcxz    ML2              ; WHILE (Cx <> 0)   (NOT EOF)

        Call    P2               ;   Udfør P2                 

        Mov     Si,Offset DataA  ;   Ds:Si = Adr. på DataA    

        Call    P1               ;   Udfør P1                 

        Jmp     ML1              ; ENDWHILE                   

ML2:    Mov     Al,0             ; Opsæt returkode 0          

        Mov     Ah,4CH           ;   og                       

        Int     21H              ;    returner til DOS        

Main    EndP                     ; Slut på hovedrutine        

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

P1      Proc    Near             ; Start på definition af P1  

        Mov     Dx,Si            ; Ds:Dx = Adr. på buffer     

        Mov     Cx,128           ; Cx = bufferlængde          

        Xor     Bx,Bx            ; Bx = 0 (std. input)        

        Mov     Ah,3Fh           ; opsæt funktion for læs     

        Int     21h              ;   og udfør.                

        Mov     Cx,Ax            ; Cx = antal læste byte      

        Ret                      ; returner fra P1            

P1      EndP                     ; Slut på definition af P1   

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

P2      Proc    Near             ; Start på definition af P2  

        Mov     Dx,Cx                                         

        Mov     Cl,4                                          

        Xor     Ah,Ah                                         

        Xor     Bh,Bh                                         

P2L1:                                                         

        Xor     Di,Di                                         

        Mov     Ch,8                                        

P2L2:                                                         

        Mov     Al,Ds:[Si]                                    

        Shl     Ax,Cl                                         

        Shr     Al,Cl                                         

        And     Ah,00001111B                                  

        Mov     Bl,Ah                                         

        Mov     Ah,DataC[Bx]                                  

        Mov     DataB[Di],Ah                                  

R1:                                                           

        Inc     Di                                            

        Mov     Bl,Al                                         

        Mov     Al,DataC[Bx]                                  

        Mov     DataB[Di],Al                                  

        Inc     Di                                            

        Inc     Di                                            

        Inc     Si                                            

        Dec     Ch                                            

R2:                                                           

        Jnz     P2L2                                          

        Mov     Bx,Dx                                         

        Mov     Dx,Offset DataB                               

        Mov     Ah,09h                                        

        Int     21h                                           

        Mov     Dx,Bx                                         

        Cmp     Si,Dx                                         

        Jl      P2L1                                          

        Ret                                                   

P2      EndP                     ; Slut på definition af P2   

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Cseg    EndS                     ; slut på code‑segment       

Dseg    Segment Para Public 'DATA'                            

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

DataA   Db      128 Dup (0)                                   

DataB   Db      '.. .. .. .. .. .. .. ..',13,10,'$'           

DataC   Db      '0123456789ABCDEF'                                      

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Dseg    EndS                     ; slut på data‑segment       

Sseg    Segment Para Stack 'STACK'                            

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

        dw      256 dup(?)       ; 256 ord sat af til stack   

;‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑‑

Sseg    EndS                     ; slut på stack‑segment      

        End     Main             ; Slut, opsæt startadresse   

